Computer Science 773

Robotics and Real-time Control

NUMERICALLY CONTROLLED TOOLS

EXAMPLE OF A MILLING SEQUENCE.

(G.Stute and H.Eitel : *The milling technology in Exapt 3*, in W.H.P.Leslie (Ed) : *NC Programming Languages* (North-Holland, 1970))

THE PART TO BE CONSTRUCTED :

THE MILLING SEQUENCE :

First :

Second :

Third :

Fourth :

EXAMPLE OF A PROGRAMME IN 2C,L.
(D.G. Wilkinson : *Cam manufacture*, in W.H.P.Leslie (Ed) :
NC Programming Languages (North-Holland, 1970))

THE SYSTEM AND SOME VARIABLES :



THE PROGRAMME :

```
PARTNO C205  CAM. (D-R-D-R-O)
REMARK   CAM HAS A ROLLER FOLLOWER MOUNTED ON A RADIUS ARM.
REMARK   D.G. WILKINSON N.E.L. X5 17-12-68.
CLPRNT

REMARK   INPUT VARIABLES FOLLOW.
XCENT  =  3.5    $$ X DISTANCE FROM CENTRE OF CAM TO CENTRE OF PIVOT.
YCENT  =  3.026  $$ Y DISTANCE FROM CENTRE OF CAM TO CENTRE OF PIVOT.
STRTRD =  2.053  $$ BASE RADIUS OF CAM.
FOLDIA =  1.084  $$ DIAMETER OF FOLLOWER
RADARM =  3.55   $$ LENGTH OF RADIUS ARM.
LMAX   =  16     $$ MAXIMUM ANGULAR TRAVEL OF RADIUS ARM.
ALFA1  =  24     $$ STARTING ANGLE OF FIRST LIST PERIOD.
THETA1 =  60     $$ ANGLE OF FIRST LIFT PERIOD.
ALFA2  =  108    $$ STARTING ANGLE OF SECOND LIST PERIOD.
THETA2 =  195    $$ ANGLE OF SECOND LIFT PERIOD.

REMARK   CUTTING VARIABLES FOLLOW.
CUTDIA =  1    $$ CUTTER DIA. IN INCHES.
SPEED1 =  400  $$ SPINDLE SPEED IN RPM.
RAPID1 =  15   $$ RAPID FEED RATE ( INS/MIN).
ROF1   =  5    $$ ROUGHING FEED RATE (INS/MIN).
FIN1   =  1    $$ FINISHING FEED RATE ( INS/MIN).
```

```
REMARK   STATIC GEOMETRY CALCULATIONS.
DIAG   =  SQRT(XCENT**2+YCENT**2) $$ RADIAL DISTANCE TO PIVOT.
TWOPI  =  3.14159  *2
TEMP1  =  RADARM**2+DIAG**2 $$ B**2 + C**2 USED IN COSINE FORMULA.
TEMP2  =  2*RADARM*DIAG     $$ 2*B*C  IN COSINE FORMULA.
MINDST =  STRTD+FOLDIA/2 $$ MINIMAL RADIAL DISTANCE TO FOLLOWER CENTRE.
ANG1   =  ATAN(YCENT/XCENT) $$ ANGLE OF PIVOT FROM POS. X-AXIS.
ANG2   =  ACOS((TEMP1-MINDST**2)/TEMP2) $$ MIN. ANGLE BETWEEN
                                        $$ RADARM AND DIAG,
MAXDST = SQRT( TEMP1-TEMP2*COS(ANG2+LMAX)) $$ MAX. RADIUS
                                           $$ TO FOLLOWER CENTRE.
ENDRAD =  MAXDST-FOLDIA/2 $$ MAXIMUM CAM RADIUS.
          PRINT/3,DIAG,MINDST,ANG1,ANG2,MAXDST,ENDRAD

REMARK   INITIALIZE VARIABLES.
THETA  =  0         $$ CUMULATIVE CAM LIFT ANGLE.
GAMMA1 =  ALFA1+ANG1-2 $$ CUMULATIVE ANGLE OF PIVOT FROM POS. X-AXIS.
GAMMA2 =  ANG2      $$ CUMULATIVE ANGLE BETWEEN RADARM AND DIAG.
DELTA1 =  THETA1/8  $$ 8 INCREMENTS ALONG LIFT CURVE.
THETAT =  THETA1    $$ TOTAL CAM LIFT ANGLE.

REMARK   GEOMETRY MACRO FOR DEFINING CENTRE OF FOLLOWER (PQ).
MACQ1  =  MACRO/PQ
RADQ   =  SQRT(TEMP1-TEMO2*COS(GAMMA2)) $$ RADIUS TO CENTRE OF FOLLOWER.
GAMMA3 =  ASIN(SIN(GAMMA2)*RADARM/RADQ) $$ ANGLE BETWEEN PIVOT AND FOLLOWER.
PQ     =  POINT/ORIG,THETAR,(GAMMA1+GAMMA3),RADQ
          TERMAC

REMARK   LIFT FUNCTION MACRO.
MACQ2  =  MACRO/
LIFT   =  LMAX*(THETA/THETAT-SIN(TEMP3*THETA)/TWOPI) $$ LIFT ANGLE OF
          TERMAC                                     $$ RADIUS ARM.

REMARK   CALLING MACRO FOR LIFT PERIOD.
MACQ3  =  MACRO/PQ1
          CALL/MACQ1,PQ=PQ1  $$ DEFINE POINT PQ.
THETA  =  THETA+DELTA1       $$ INCREMENT CAM LIFT ANGLE.
          CALL/MACQ2         $$ CALCULATE LIFT ANGLE OF RADIUS ARM.
GAMMA1 =  GAMMA1+ABS(DELTA1) $$ INCREMENT ANGLE OF PIVOT.
GAMMA2 =  ANG2+LIFT          $$ INCREMENT ANGLE BETWEEN RADARM AND DIAG.
          TERMAC

REMARK   MACRO TO DEFINE FOLLOWER CURVE AND CAM PROFILE.
MACQ4  =  MACRO/TABQ1,LQ1,LQ2,LQ3
TEMP3  =  360/THETAT $$ USED IN LIFT FUNCTION.
          CALL/MACQ1,PQ=P1  $$ DEFINE P1 AND P2 AT CONSTANT RADIUS.
          GAMMA1=GAMMA1+1   $$ SPACED AT 1 DEGREE INCREMENTS BEFORE
          CALL/MACQ1,PQ=P2  $$ START OF LIFT PERIOD TO GIVE CONTINUITY
          GAMMA1=GAMMA1+1   $$ OF LIFT CURVE.
          CALL/MACQ3,PQ1=P3
          CALL/MACQ3,PQ1=P4
          CALL/MACQ3,PQ1=P5
          CALL/MACQ3,PQ1=P6
          CALL/MACQ3,PQ1=P7
          CALL/MACQ3,PQ1=P8
          CALL/MACQ3,PQ1=P9
          CALL/MACQ3,PQ1=P10
          CALL/MACQ1,PQ=P11
          GAMMA1=GAMMA1+1
          CALL/MACQ1,PQ=P12
          GAMMA1=GAMMA1+1
          CALL/MACQ1,PQ=P13
TABQ   =  TABCYL/NOZ,SPLINE,P1,P2,P3,P4,P5,P6,P7,P8,P9,P10,P11,P12,P13
TABQ1  =  TABCYL/PARLEL,LEFT,TABQ,(FOLDIA/2)
LQ1    =  LINE/ORIG,P3
```

```
LQ2    =  LINE/ORIG,P11
LQ3    =  LINE/ORIG,P7
          TERMAC

REMARK    DEFIND CAM PROFILE FOR PERIOD LIFT1.
ORIG   =  POINT/0,0  $$ CENTRE OF CAM.
          CALL/MACQ4,TABQ1=LIFT1,LQ1=L1,LQ2=L2,LQ3=L1A

REMARK    DEFINE LINE GEOMETRY FOR START AND END OF CUT.
PCENT  =  POINT/ORIG,THETAR,(GAMM1+GAMMA3-2),(ENDRAD+CUTDIA*2)
BLEND  =  CIRCLE/CENTRE,PCENT,RADIUS,(CUTDIA/2)
STRTLN =  LINE/PCENT,PERPTO,L2

REMARK    RESET VARIABLES FOR SECOND LIFT PERIOD.
THETA  =  THETA2
GAMMA1 =  ALFA2+ANG1-2
GAMMA2 =  ANG2+LMAX
DELTA1 =  -THETA2/8 $$ GIVES NEGATIVE INCREMENTS.
THETAT =  THETA2

REMARK    DEFINE CAM PROFILE FOR PERIOD LIFT2.
          CALL/MACQ4,TABQ1=LIFT2,LQ1=L3,LQ2=L4,LQ3=L3A

REMARK    DEFINE REST OF GEOMETRY.
C1     =  CIRCLE/0,0,STRTRD  $$ BASE CIRCLE OF CAM.
C2     =  CIRCLE/0,0,ENDRAD  $$ MAX. CAM RADIUS CIRCLE.
SETPNT =  POINT/0,0,3   $$ SET POINT 3 INCHES ABOVE CAM CENTRE.

REMARK    MACRO FOR CUTTING ONCE ROUND THE CAM.
MACQ5  =  MACRO/FEED1
          CUTTER/KUT
          GO/ON,STRTLN,(PLANE/0,0,1,3),ON,L2,RAPID1
          GODLTA/-3
          AUTOPS
          GORGT/STRTLN,TO,BLEND
          TLRGT
          GORGT/BLEND,ON,L2,FEED1
          GOFWD/C2,ON,L3
          GOFWD/LIFT2,ON,L3A
          GOFWD/LIFT2,ON,L4
          GOFWD/C1,ON,L1
          GOFWD/LIFT1,ON,L1A
          GOFWD/LIFT1,ON,L2
          GOFWD/BLEND,ON,STRTLN
          GODLTA/3,RAPID1
          GOTO/SETPNT
          TERMAC

          SPINDL/SPEED1
PPRINT    ROUGH OUT CAM .02 INCH OVERSIZE.
          FROM/SETPNT
          INTOL/.005
          OUTTOL/.005
KUT    =  CUTDIA+.04
          CALL/MACQ5,FEED1=RUF1
          STOP
          SPINDL/SPEED1
PPRINT    FINISH CAM.
          INTOL/.00005
          OUTTOL/.00005
KUT    =  CUTDIA
          CALL/MACQ5,FEED1=FIN1
          END
          FINI
```

THE VALUE OF NUMERICAL CONTROL.

(P. Fiorellino : *Introductory course in the numerical control of machine tools* (ACIERA, 1980).)

Comparing manual control with numerical control; economy of scale.

Advantage of a "high-level" language.

DESCENDANTS OF NUMERICAL CONTROL.

(P. Fiorellino : *Introductory course in the numerical control of machine tools* (ACIERA, 1980).)

<table>
<tr><td></td><td>Degree of integration</td><td></td></tr>
<tr><td>Automated machining</td><td>Automated machining sequence</td><td>Automated manufacturing sequence</td></tr>
<tr><td>Includes
Machining with
- positioning
- measuring

NC M</td><td rowspan="2">Includes
Machining with
- positioning
- measuring

tool exchange</td><td rowspan="2">Includes
Machining with
- positioning
- measuring

tool exchange

Workpiece change

Storage and transfer

- workpiece
(-tools)

Total control
-flow of information
- flow of material</td></tr>
<tr><td>NC MC (machining centers)</td></tr>
<tr><td colspan="3">NC MC (FMS) (flexible manufacturing systems)</td></tr>
</table>

Alan Creak,
March, 1998.